user
A pangram is a sentence that contains all the letters of the English alphabet at least once.

For example:   *The quick brown fox jumps over the lazy dog*.

Task

Write a function or method to check a sentence to see if it is a   pangram   (or not)   and show its use.

Related tasks
:   * determine if a string has all the same characters
    * determine if a string has all unique characters
assistant
(formerly Perl 6)

```
constant Eng = set 'a' .. 'z';
constant Cyr = (set 'а' .. 'ё') (-) (set 'ъ', 'ѐ');
constant Hex = set 'a' .. 'f';

sub pangram($str, Set $alpha = Eng) {
  $alpha ⊆ $str.lc.comb
}

say pangram("The quick brown fox jumps over the lazy dog.");
say pangram("My dog has fleas.");
say pangram("My dog has fleas.", Hex);
say pangram("My dog backs fleas.", Hex);
say pangram "Съешь же ещё этих мягких французских булок, да выпей чаю", Cyr;

```

Output:

```
True
False
False
True
True
```